

# MIPSI: Rigid-body pendulum


## Modeling (assumptions)

The figure to the right shows a non-uniform density rigid rod  $A$  attached by a frictionless revolute/pin joint to Earth  $N$  (a Newtonian reference frame). The only external forces on rod  $A$  are contact forces at  $N_o$  (the point of  $N$  in contact with  $A$ ) and Earth's gravitational forces (modeled as local and uniform).

## Identifiers (label $L$ , $\theta$ , and unit vectors $\hat{n}_x, \hat{n}_y, \hat{n}_z, \hat{a}_x$ on the figure)

Right-handed orthogonal unit vectors  $\hat{n}_x, \hat{n}_y, \hat{n}_z$  and  $\hat{a}_x, \hat{a}_y, \hat{a}_z$  are fixed in  $N$  and  $A$ , respectively, with  $\hat{n}_x$  horizontally-right,  $\hat{n}_y$  vertically-upward,  $\hat{a}_y$  directed from  $A_{cm}$  ( $A$ 's center of mass) to  $N_o$ , and  $\hat{a}_z = \hat{n}_z$  parallel to the pin.

Description	Symbol	Type	Value
Earth's gravitational constant	$g$	Constant	9.8 m/s <sup>2</sup>
Distance between $N_o$ and $A_{cm}$	$L$	Constant	7.5 cm
Mass of rod $A$	$m$	Constant	10 grams
$A$ 's moment of inertia about $A_{cm}$ for $\hat{a}_z$	$I$	Constant	50 g * cm <sup>2</sup>
Angle from $\hat{n}_y$ to $\hat{a}_y$ with $+\hat{n}_z$ sense	$\theta$	Variable	




## Physics

Answers at [www.MotionGenesis.com](http://www.MotionGenesis.com) ⇒ [Get Started](#) ⇒ Pendulum.

Complete rod  $A$ 's *free-body diagram (FBD)* (to the right).

Form an **ODE** (ordinary differential equation) governing this pendulum's motion.




$$(I + mL^2)\ddot{\theta} + mgL \sin(\theta) = 0 \quad \Rightarrow \quad \ddot{\theta} + \frac{mgL}{I + mL^2} \sin(\theta) = 0$$

## Simplify and solve

By defining the constant  $\omega$  (shown right), the previous ODE is written in a form that clearly shows it is **nonlinear** due to  $\sin(\theta)$ .

Since the ODE is **nonlinear**, a closed-form (analytical/symbolic) solution is difficult. The plot shows a numerical (computer) solution to this nonlinear ODE when the pendulum is released from rest with  $\theta(t=0) = 90^\circ$  or  $\theta(t=0) = 30^\circ$ .




Alternately, making the small-angle approximation  $\sin(\theta) \approx \theta$  produces the **linear** ODE:

$$\ddot{\theta} + \omega^2 \theta \approx 0 \quad \text{whose closed-form (analytical/symbolic) solution is } \boxed{\theta(t) \approx \theta(0) \cos(\omega t)}.$$

## Interpret (communicate)

The pendulum swings back and forth with an amplitude that depends on starting angle (when released from  $90^\circ$ , it swings higher than when released from  $30^\circ$ ). The solution to the **linear** ODE says the pendulum's oscillation frequency  $\omega$  does not depend on release angle. However, the plot of the **nonlinear** ODE shows the pendulum has less-frequent oscillations when released from  $90^\circ$  than when released from  $30^\circ$ . This is due to the linear ODE only approximating the nonlinear ODE accurately at small angles ( $\theta \ll 1$  radians).

```
% File: RigidBodyPendulumDynamics.txt
% Copyright (c) 2009 Motion Genesis LLC. All rights reserved.
%-----
NewtonianFrame N          % Earth.
RigidBody       A          % Rod.
%-----
Variable  theta''        % Angle and 1st/2nd time-derivatives.
Constant  g = 9.8 m/s^2
Constant  L = 7.5 cm
A.SetMassInertia( m = 10 grams, IAx, IAY, I = 50 g*cm^2 )
%-----
%           Rotational and translational kinematics.
A.RotateZ( N, theta )
Acm.Translate( No, -L*Ay> )
%-----
%           Add relevant forces.
Acm.AddForce( -m*g*Ny> )
%-----
%           Rotational equation of motion.
Dynamics = Dot( Nz>, A.GetDynamics(No) )
Solve( Dynamics, theta'' )
%-----
%           Integration parameters and initial values.
Input  tFinal = 4 sec, tStep = 0.02 sec, absError = 1.0E-07
Input  theta = 90 deg, theta' = 0.0 rad/sec
%-----
%           List output quantities and solve ODEs.
OutputPlot t seconds, theta degrees
ODE() RigidBodyPendulumDynamics
%-----
%           Record input together with responses
Save RigidBodyPendulumDynamics.all
Quit
```



<sup>3</sup>The Shortt clock pictured here is the most accurate mechanical clock ever made, with an error rate of 1 sec in 12 years. The clock is so sensitive it detects tiny changes in gravity due to tidal distortions on Earth caused by the Sun and Moon's gravity. Paul Heyl from NIST used it for a new determination of the gravitational constant.